European Wolfram Technology Conference 2017

Computer Assisted Physics at Twente University

Implementation of Mathematica in the Applied Physics Curriculum

Dr. J.W.J. Verschuur
University of Twente; the Netherlands

Contents

- Motivation
- Goal in terms of end terms
- Design in the curriculum
- Items addressed:
  - skills → structure, syntax and punctuation
  - concepts →
  - Application and integration → Physics
- Realization
- Experience so far
- Items for discussion

Motivation

Motivation in line with the message of Conrad Wolfram's TED-talk on July 2010:



- Emphasis on translation of physics problem to mathematics
- Transfer hand-head calculation to Computer → Mathematica; don't get trapped in sub-routines!
- Emphasis on interpretation of mathematics results in terms of Physics

Additional to that:

- Learn Students to work with tools, but also
- Learn Students to think in using the possibilities → innovation

Applied Physics Curriculum - 1st year

1. Calculus 1, mechanical dynamics, experiments, data collection and presentation
2. Calculus 2, thermo-dynamics, experiments, data analysis
3. Linear Algebra, Quantum matter, geometrical optics + experiments, Mathematica, with application in optical system design.
4. Vector Calculus, Electrodynamics, Mathematica workshop on vector calculus and concepts of Electrodynamics

Applied Physics Curriculum - 2nd year

1. Systems, models and signals → Fourier, Laplace, z-Transform, Simulations
2. Quantum Mechanics, Optics, Hilbert spaces
3. Statistical Physics, Solid-State Physics, Partial Differential Equations
4. Electrodynamics, Fluid Physics, Numerical Methods

Support by the CAPh-project

1. M3: Introduction and application in optics
2. M4: extension of the skills and application in Electrodynamics
3. M5: Signals, Models and Systems: only passive use in lectures
4. M6: Quantum Mechanics: in preparation

Zero Measurement

Before introduction to Mathematica:

Use of Wolfram Alpha:

- 50% of the students uses Wolfram Alpha 0-4 times a week
- 19% of the students uses Wolfram Alpha >4 times a week
- 27% prefers pen and paper
- 4% already uses Mathematica

Expectation on focus shift in Mathematics:

- 67% expects shift from calculation to concepts
- 5% does not expect a change
- 28% has no idea

Status October 2017:

Realized activities in first year Applied Physics

- Introductory workshops: 3 sessions.
- Application assignment: use matrix multiplication to design an optical system with the ABDC-formalism (microscope, telescope or zoom lens).
- Advanced workshop: vector fields and functions, multiple integrals and visualization.
- Application case in Electrostatics using vector calculus and vector theorems.

Student's perspective experience

- Why do I need to learn an other mathematical software package? I know already ...
- Worksheets start-off too simple, but become difficult too fast.
- Students recognize the potential advantage, however are not convinced yet.
- A choice at the exam between Mathematica and pen & paper is regarded as "not fair". This is remarkable, but ...
- You don't have to be an addict to use a computer; however some students just don't like it at all.

Teacher's perspective experience

- Although goals and process of the session are outlined, Student's expectations are hard to get right.
- Pay attention to the investment students have to make.
- Balance the learning of skills with useful applications that pay off; like doing sports or making music.
- Pay attention to the enormous diversity in motivation of the students → common interest is Physics.

Balance skills training ↔ applications

You have to know the philosophy & machinery

- You should define everything explicitly. No implicit conditions from context.
- You should know some of the mechanics behind the curtain, e.g. "=" assignments, comparison.
- You should know that Mathematica calculates exact.
- You should know that syntax has a meaning; and it makes sense.
- You should know the mathematics behind the calculations
- Do not redo the "pen and paper" mathematics -> misses the goal, frustrating.
- etc ...

Illustration

A few examples

In[1]:= $\{\text{Sin}[1], \text{Sin}[\frac{\pi}{2}], \text{N}[\text{Sin}[1]]\}$

```
Out[1]= {Sin[1], 1, 0.841471}
```

In[2]:= f[x_] = a x^2 + b x + c

Out[2]= $c + b x + a x^2$

In[3]:= parval = {a → 2, b → 5, c → 1}

Out[3]= {a → 2, b → 5, c → 1}

In[4]:= Solve[f[x] == 0, x]

Out[4]= $\left\{\left\{x \to \frac{-b - \sqrt{b^2 - 4 a c}}{2 a}\right\}, \left\{x \to \frac{-b + \sqrt{b^2 - 4 a c}}{2 a}\right\}\right\}$

In[5]:= sol1 = Solve[(f[x] /. parval) == 0, x]

Out[5]= $\left\{\left\{x \to \frac{1}{4}\left(-5 - \sqrt{17}\right)\right\}, \left\{x \to \frac{1}{4}\left(-5 + \sqrt{17}\right)\right\}\right\}$

In[6]:= N[sol1]

Out[6]= {{x → -2.28078}, {x → -0.219224}}

Introduction example

Use a few examples from the introduction workshops (e.g. the v.d. Waals equation)

In[7]:= $Assumptions = {a, b, nn, k, T, v, p, tp, t, θ, ϕ, ℏ, ϵ0} ∈ Reals ∧
T > 0 ∧ k > 0 ∧ v > 0 ∧ a > 0 ∧ b > 0 ∧ nn > 0 ∧ ℏ > 0 ∧ ϵ0 > 0 ∧ t > 0 ∧ tp > 0 ∧ ℏ > 0 ∧ {l, m} ∈ Integers;

In[8]:= imgSize = {600, 400}

Out[8]= {600, 400}

In[9]:= $p1[v\_, t\_] = \frac{nn\, k\, t}{v - nn\, b} - a \frac{nn^2}{v^2}$

Out[9]= $-\frac{a\, nn^2}{v^2} + \frac{k\, nn\, t}{-b\, nn + v}$

In[10]:= v12 = Solve[{∂_v p1[v, t] == 0, $\partial_{v,v}$ p1[v, t] == 0}, {v, t}, Reals] // FullSimplify

Out[10]= $\left\{\left\{v \to 3\, b\, nn, t \to \frac{8 a}{27 b k}\right\}\right\}$

In[11]:= critp = {vc = v /. v12[[1]], pc = p1[v, t] /. v12[[1]], Tc = t /. v12[[1]]}

Out[11]= $\left\{3\, b\, nn, \frac{a}{27 b^2}, \frac{8 a}{27 b k}\right\}$

In[12]:= $ps[x\_] = Normal\left[Series\left[p1\left[\frac{nn}{x}, t\right], \{x, 0, 3\}\right]\right]$

Out[12]= $k\, t\, x + (-a + b\, k\, t)\, x^2 + b^2\, k\, t\, x^3$

In[13]:= $pss[v\_] = ps\left[\frac{nn}{v}\right]$

Out[13]= $\frac{b^2\, k\, nn^3\, t}{v^3} + \frac{nn^2\, (-a + b\, k\, t)}{v^2} + \frac{k\, nn\, t}{v}$

Visualized Van Der Waals curves with Temperature as parameter.

In[14]:= cd1 = {a → 0.01, b → 0.01, k → 1, nn → 10}

Out[14]= {a → 0.01, b → 0.01, k → 1, nn → 10}

In[15]:=
```
Plot[{p1[v, 0.27], p1[v, 0.28], p1[v, 0.29], p1[v, 0.30]} /. cd1, {v, 0.1, 1},
 AxesLabel → {"V", "P"}, PlotStyle → {Red}, BaseStyle → {FontSize → 24}, ImageSize → imgSize]
```

Out[15]=



In[16]:=
```
p3[v_, t_] = p1[v, t] /. cd1; critp3 = critp[[2]] /. cd1;
```

In[17]:=
```
Manipulate[Plot[{p3[v, tp], critp3}, {v, 0.1, 1}, AxesLabel → {"V", "P(V;T)"}, PlotRange → {2, 7},
  PlotStyle → {Red, {Blue, Dotted}}, BaseStyle → {FontSize → 24}, ImageSize → imgSize], {{tp, 0.28, "Temperature"}, 0.20, 0.35}]
```

Out[17]=



In[18]:=
```
critp /. cd1
```

Out[18]= {0.3, 3.7037, 0.296296}

Design assignment of optical system

Using the ABCD-matrix formalism and linear algebra → design an optical system of a few lenses.
Use Manipulate of various parameters to get a “feeling” on the effectiveness of the various parameters.

Subjects:

- Microscope
- Telescope
- Zoom-lens

Electrodynamics example



Electrodynamics Workshop

use case of vector calculus, with the given E-field, flux charge distribution, vector theorem, etc.

Electric field → Charge Distribution

Given is the following Electric field:

```
In[19]:=  ee[x_, y_] = {y^2 x, y x^2, 0}

Out[19]=  {x y^2, x^2 y, 0}
```

Plot the vector Field in 3D

In[20]:=
```
SliceVectorPlot3D[ee[x, y], {z == 0, z == 1/2, z == -1/2}, {x, -2, 2}, {y, -2, 2},
  {z, -1, 1}, BaseStyle → {FontSize → 24}, ImageSize → imgSize, AxesLabel → {"x", "y", "z"}]
```

Out[20]=



Top view of the x-y plane: make a VectorPlot of the field at z=0 in 2D

In[21]:=
```
vecp = VectorPlot[ee[x, y].{1, 0, 0} {1, 0} + ee[x, y].{0, 1, 0} {0, 1}, {x, -2, 2},
  {y, -2, 2}, VectorStyle → Red, FrameLabel → {"x", "y"}, BaseStyle → {FontSize → 24}, ImageSize → imgSize]
```

Out[21]=



Determine the Curl of the field

This should be equal to zero for a valid Electric field. (use <exc>del<esc><ctl>_ , with the appropriate coordinate system variables and name indication.

```
In[22]:= ∇{x,y,z},"Cartesian" × ee[x, y]

Out[22]= {0, 0, 0}
```

Calculate the corresponding charge density

Calculate the charge density as a function of the coordinates from the electric field.

```
In[23]:= ρ[x_, y_, z_] = ϵ0 ∇{x,y,z},"Cartesian" .ee[x, y]
```

Out[23]= $(x^2 + y^2)\,\epsilon 0$

Plot the charge distribution (use a ContourPlot in 2D (x,y), since the distribution is not dependent on z). Use as a condition that ϵ0 →1, and use PlotLegend and FrameLabel.

```
In[24]:= chargep = ContourPlot[ρ[x, y, 0] /. ϵ0 → 1, {x, -2, 2}, {y, -2, 2},
           PlotLegends → Automatic, FrameLabel → {"x", "y"}, BaseStyle → {FontSize → 24}, ImageSize → imgSize]
```

Out[24]=



Show charge distribution and Field vectors in a single graph:

In[25]:= `Show[{chargep, vecp}]`

Out[25]=



Check Gauss' theorem in a cube

Calculate the flux through the 6 surfaces of a cube with dimension *a* by *a* by *a*.

Integrate over surface in the x-z-direction:

In[26]:= `inty =` $\int_0^a \int_0^a$ `ee[x, y].{0, 1, 0}` $dx\,dz$

Out[26]= $\frac{a^4\, y}{3}$

Integrate over surface in the y-z-direction:

In[27]:= `intx =` $\int_0^a \int_0^a$ `ee[x, y].{1, 0, 0}` $dy\,dz$

Out[27]= $\frac{a^4\, x}{3}$

Integrate over surface in the x-y-direction: (should be zero, no component of the field in the z-direction)

In[28]:= `intz =` $\int_0^a \int_0^a$ `ee[x, y].{0, 0, 1}` $dx\,dy$

Out[28]= 0

Calculate the total flux integral

In[29]:= `inttot = (intx /. x → 0) + (intx /. x → a) + (inty /. y → 0) + (inty /. y → a) + (intz /. z → 0) + (intz /. z → a)`

Out[29]= $\frac{2\, a^5}{3}$

Calculate the total charge in the cube from the charge density

In[30]:= $\text{Qtot} = \int_0^a \int_0^a \int_0^a \rho[x, y, z] \, dx \, dy \, dz$

Out[30]= $\frac{2\, a^5\, \epsilon 0}{3}$

Check the relation between nett flux and charge:

In[31]:= $\text{inttot} == \frac{\text{Qtot}}{\epsilon 0}$

Out[31]= True

Check Conservativeness of field via path integrals

Calculate the potential in point {*a*,*a*,*a*} given that the potential in {0,0,0} is zero.

First path: start along the x-axis, then parallel to the y-axis and finish parallel to the z-axis:

In[32]:= $\text{Vaaa1} = -\int_0^a \text{ee}[x, 0].\{1, 0, 0\} \, dx - \int_0^a \text{ee}[a, y].\{0, 1, 0\} \, dy - \int_0^a \text{ee}[a, a].\{0, 0, 1\} \, dz$

Out[32]= $-\frac{a^4}{2}$

Second path: start along the y-axis, then parallel to the x-axis and finish parallel to the z-axis:

In[33]:= $\text{Vaaa2} = -\int_0^a \text{ee}[0, y].\{0, 1, 0\} \, dy - \int_0^a \text{ee}[x, a].\{1, 0, 0\} \, dx - \int_0^a \text{ee}[a, a].\{0, 0, 1\} \, dz$

Out[33]= $-\frac{a^4}{2}$

Compare the potential calculation along the two different paths

In[34]:= `Vaaa1 == Vaaa2`

Out[34]= True

Both paths, give the same value, as it should be and consistent with the fact that the Curl of the field is equal to zero.

Extra Problems

- Repeat the above calculations for the three vector fields given below, as far as these fields are real vector fields.

1. Field 2 in Cartesian coordinates:

In[35]:= `ee2[x_, y_, z_] = k {x y, 2 y z, 3 x z}`

Out[35]= `{k x y, 2 k y z, 3 k x z}`

First check whether this could be an Electric field

In[36]:= $\nabla_{\{x,y,z\},\text{"Cartesian"}} \times$ ee2[x, y, z]

Out[36]= {-2 k y, -3 k z, -k x}

This is not an Electric field.

2. Field 3 in Cartesian coordinates:

In[37]:= ee3[x_, y_, z_] = k $\{y^2, 2\,x\,y + z^2, 2\,y\,z\}$

Out[37]= $\{k\,y^2, k\,(2\,x\,y + z^2), 2\,k\,y\,z\}$

In[38]:= $\nabla_{\{x,y,z\},\text{"Cartesian"}} \times$ ee3[x, y, z]

Out[38]= {0, 0, 0}

This is an Electric Field.

```
In[39]:= SliceVectorPlot3D[ee3[x, y, z] /. k → 1, "Midplanes", {x, -2, 2}, {y, -2, 2},
           {z, -2, 2}, AxesLabel → {"x", "y", "z"}, BaseStyle → {FontSize → 24}, ImageSize → imgSize]
```

SliceVectorPlot3D: Slice specification Midplanes should be a named slice, equation, surface or volume region, or list of slices.

Out[39]=



In[40]:=
```
SliceVectorPlot3D[ee3[x, y, z] /. k → 1, {z == -1, z == 0, z == 1, "BackPlanes"}, {x, -2, 2},
  {y, -2, 2}, {z, -2, 2}, AxesLabel → {"x", "y", "z"}, BaseStyle → {FontSize → 24}, ImageSize → imgSize]
```

Out[40]=



In[41]:=
```
ρ3[x_, y_, z_] = ϵ0 ∇{x,y,z},"Cartesian" .ee3[x, y, z]
```

Out[41]= $(2\,k\,x + 2\,k\,y)\,\epsilon 0$

In[42]:=
```
chargep3 = ContourPlot[ρ3[x, y, 0] /. {ϵ0 → 1, k → 1}, {x, -2, 2}, {y, -2, 2},
  PlotLegends → Automatic, FrameLabel → {"x", "y"}, BaseStyle → {FontSize → 24}, ImageSize → imgSize]
```

Out[42]=



Calculate the flux through the 6 surfaces of a cube with dimension *a* by *a* by *a*.

in the y direction:

In[43]:= $\text{inty3} = \int_0^a \int_0^a \text{ee3}[x, y, z].\{0, 1, 0\}\, dx\, dz$

Out[43]= $\frac{1}{3} a^3 k (a + 3 y)$

In the x-direction:

In[44]:= $\text{intx3} = \int_0^a \int_0^a \text{ee3}[x, y, z].\{1, 0, 0\}\, dy\, dz$

Out[44]= $\frac{a^4 k}{3}$

in the z-direction: (should be zero, no component of the field in the z-direction)

In[45]:= $\text{intz3} = \int_0^a \int_0^a \text{ee3}[x, y, z].\{0, 0, 1\}\, dx\, dy$

Out[45]= $a^3 k z$

Calculate the total flux integral

In[46]:= `inttot3 = -(intx3 /. x → 0) + (intx3 /. x → a) - (inty3 /. y → 0) + (inty3 /. y → a) - (intz3 /. z → 0) + (intz3 /. z → a)`

Out[46]= $2 a^4 k$

Calculate the total charge in the cube from the charge density

In[47]:= $\text{Qtot3} = \int_0^a \int_0^a \int_0^a \rho 3[x, y, z]\, dx\, dy\, dz$

Out[47]= $2 a^4 k \epsilon 0$

Check the relation between flux and charge:

In[48]:= $\text{inttot3} == \frac{\text{Qtot3}}{\epsilon 0}$

Out[48]= True

Calculate the potential in point {*a*,*a*,*a*} given that the potential in {0,0,0} is zero.

first path, along the x-axis, parallel to the y-axis, parallel to the z-axis

In[49]:= $\text{V3aaa1} = -\int_0^a \text{ee3}[x, 0, 0].\{1, 0, 0\}\, dx - \int_0^a \text{ee3}[a, y, 0].\{0, 1, 0\}\, dy - \int_0^a \text{ee3}[a, a, z].\{0, 0, 1\}\, dz$

Out[49]= $-2 a^3 k$

second path : along the y-axis, parallel to the x-axis, parallel to the z-axis

In[50]:= $\text{V3aaa2} = -\int_0^a \text{ee3}[0, y, 0].\{0, 1, 0\}\, dy - \int_0^a \text{ee3}[x, a, z].\{1, 0, 0\}\, dx - \int_0^a \text{ee3}[a, a, z].\{0, 0, 1\}\, dz$

Out[50]= $-2 a^3 k$

Compare the potential calculation along the two different paths

In[51]:= `V3aaa1 == V3aaa2`

Out[51]= True

Both paths, give the same value, as it should be.

3. Field 4 in Spherical coordinates. For this case substitute a sphere of radius a for the flux calculations and the total charge calculation. For the calculation of the potential, chose two different paths between two points with different values of all three coordinates {r, θ, ϕ}

In[52]:= `ee4[r_, θ_, ϕ_] = k/r {3, 2 Sin[θ] Cos[θ] Sin[ϕ], Sin[θ] Cos[ϕ]}`

Out[52]= $\left\{\frac{3\,k}{r}, \frac{1}{r}\, -2\,k\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[\phi], \frac{k\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta]}{r}\right\}$

In[53]:= `∇{r,θ,ϕ},"Spherical" × ee4[r, θ, ϕ]`

Out[53]= {0, 0, 0}

So this is an Electric Field.

Before Plotting the field first has to be converted to a Cartesian coordinate system.

In[54]:= `ee4c[x_, y_, z_] = TransformedField["Spherical" → "Cartesian", ee4[r, θ, ϕ], {r, θ, ϕ} → {x, y, z}] // Simplify`

Out[54]= $\left\{\left(k\,x\left(\frac{2\,y\,z^2}{\sqrt{x^2+y^2}}+3\left(x^2+y^2+z^2\right)-\frac{y\left(x^2+y^2+z^2\right)}{\sqrt{x^2+y^2}}\right)\right)\Big/\left(x^2+y^2+z^2\right)^2,\right.$
$\left(k\left(\frac{2\,y^2\,z^2}{\sqrt{x^2+y^2}}+3\,y\left(x^2+y^2+z^2\right)+\frac{x^2\left(x^2+y^2+z^2\right)}{\sqrt{x^2+y^2}}\right)\right)\Big/\left(x^2+y^2+z^2\right)^2, \left.\left(k\,z\left(3\,x^2+3\,y^2-2\,y\sqrt{x^2+y^2}+3\,z^2\right)\right)\Big/\left(x^2+y^2+z^2\right)^2\right\}$

In[55]:= `SliceVectorPlot3D[ee4c[x, y, z] /. k → 1, {"CenterSphere", "BackPlanes"}, {x, -2, 2}, {y, -2, 2}, {z, -2, 2}, AxesLabel → {"x", "y", "z"}, BaseStyle → {FontSize → 24}, ImageSize → imgSize]`

Out[55]=



In[56]:= `{{1, 0, 0}, {0, 1, 0}}.ee4c[x, y, z] /. k → 1`

Out[56]= $\left\{\left(x\left(\frac{2\,y\,z^2}{\sqrt{x^2+y^2}}+3\left(x^2+y^2+z^2\right)-\frac{y\left(x^2+y^2+z^2\right)}{\sqrt{x^2+y^2}}\right)\right)\Big/\left(x^2+y^2+z^2\right)^2, \left(\frac{2\,y^2\,z^2}{\sqrt{x^2+y^2}}+3\,y\left(x^2+y^2+z^2\right)+\frac{x^2\left(x^2+y^2+z^2\right)}{\sqrt{x^2+y^2}}\right)\Big/\left(x^2+y^2+z^2\right)^2\right\}$

In[57]:=
```
Manipulate[VectorPlot[{{1, 0, 0}, {0, 1, 0}}.ee4c[x, y, zz] /. k → 1, {x, -2, 2}, {y, -2, 2},
  VectorStyle → Red, FrameLabel → {"x", "y"}, BaseStyle → {FontSize → 24}, ImageSize → imgSize], {zz, -2, 2}]
```

Out[57]=



Calculate the charge density

In[58]:= $\rho 4[r\_, \theta\_, \phi\_] = \epsilon 0\, \nabla_{\{r,\theta,\phi\},\text{"Spherical"}} \cdot ee4[r, \theta, \phi]$ // FullSimplify

Out[58]= $\frac{1}{r^2} 3\, k\, \epsilon 0\, (1 + \text{Cos}[2\, \theta]\, \text{Sin}[\phi])$

Total flux through sphere of radius a

In[59]:= $\text{flux4} = \int_0^{2\pi}\int_0^{\pi} ee4[a, \theta, \phi].\{1, 0, 0\}\, a^2\, \text{Sin}[\theta]\, d\theta\, d\phi$

Out[59]= $12\, a\, k\, \pi$

Total charge in sphere of radius a

In[60]:= $\text{Qtot4} = \int_0^{2\pi}\int_0^{\pi}\int_0^{a} \rho 4[r, \theta, \phi]\, r^2\, \text{Sin}[\theta]\, dr\, d\theta\, d\phi$

Out[60]= $12\, a\, k\, \pi\, \epsilon 0$

In[61]:= $\text{flux4} == \frac{\text{Qtot4}}{\epsilon 0}$

Out[61]= True

Calculate the potential between two points along two different trajectories

- First path: straight line from $\left\{\frac{a}{2}, 0, 0\right\} \to \{a,0,0\}$, followed by a quart circle from $\{a,0,0\} \to \left\{\frac{\sqrt{2}}{2} a, 0, \frac{\sqrt{2}}{2} a\right\}$, followed by a half circle from $\left\{\frac{\sqrt{2}}{2} a, 0, \frac{\sqrt{2}}{2} a\right\} \to \left\{0, \frac{\sqrt{2}}{2} a, \frac{\sqrt{2}}{2} a\right\}$

In[62]:= $\text{V4aa1} = -\int_{\frac{a}{2}}^{a} \text{ee4}\left[r, \frac{\pi}{2}, 0\right].\{1, 0, 0\}\, dr - \int_{\frac{\pi}{2}}^{\frac{\pi}{4}} \text{ee4}[a, \theta, 0].\{0, 1, 0\}\, a\, d\theta - \int_{0}^{\frac{\pi}{2}} \text{ee4}\left[a, \frac{\pi}{4}, \phi\right].\{0, 0, 1\}\, a \,\text{Sin}\left[\frac{\pi}{4}\right] d\phi$

Out[62]= $-\frac{k}{2} - k\, \text{Log}[8]$

- Second path: half circle from $\left\{\frac{a}{2}, 0, 0\right\} \to \left\{0, \frac{a}{2}, 0\right\}$, followed by a quart circle from $\left\{0, \frac{a}{2}, 0\right\} \to \left\{0, \frac{\sqrt{2}}{4} a, \frac{\sqrt{2}}{4} a\right\}$, followed by a straight line from $\left\{0, \frac{\sqrt{2}}{4} a, \frac{\sqrt{2}}{4} a\right\} \to \left\{0, \frac{\sqrt{2}}{2} a, \frac{\sqrt{2}}{2} a\right\}$.

In[63]:= $\text{V4aa2} = -\int_{0}^{\frac{\pi}{2}} \text{ee4}\left[\frac{a}{2}, \frac{\pi}{2}, \phi\right].\{0, 0, 1\}\, \frac{a}{2} \,\text{Sin}\left[\frac{\pi}{2}\right] d\phi - \int_{\frac{\pi}{2}}^{\frac{\pi}{4}} \text{ee4}\left[\frac{a}{2}, \theta, \frac{\pi}{2}\right].\{0, 1, 0\}\, \frac{a}{2}\, d\theta - \int_{\frac{a}{2}}^{a} \text{ee4}\left[r, \frac{\pi}{4}, \frac{\pi}{2}\right].\{1, 0, 0\}\, dr$

Out[63]= $-\frac{k}{2} - k\, \text{Log}[8]$

In[64]:= `V4aa1 == V4aa2`

Out[64]= `True`

- Both path integrals give the same results, as it should be.

Quantum example: Operator Commutation



Commutation in Quantum Mechanics: position and momentum

Define the position operator:

In[65]:= `ox = (x #) &`

Out[65]= `x #1 &`

Define the momentum operator:

In[66]:= $\text{op} = \left(\frac{\hbar}{i} \partial_x \#\right) \&$

Out[66]= $\frac{\hbar\, \partial_x \#1}{i} \&$

Construct the commutation relation:

```
In[67]:= xpc = (op[ox[#]] - ox[op[#]]) &

Out[67]= op[ox[#1]] - ox[op[#1]] &
```

Apply on an arbitrary function:

```
In[68]:= xpc[h[x]] // FullSimplify

Out[68]= -i ℏ h[x]
```

Quantum example: Angular momentum



Quantum example: Angular momentum

Define the Angular Momentum Operators

```
In[69]:= Lz = ℏ/i (∂_ϕ #) &

Out[69]= (ℏ ∂_ϕ #1)/i &

In[70]:= Lp = (ℏ e^(+i ϕ) (∂_θ # + i Cos[θ]/Sin[θ] ∂_ϕ #)) &

Out[70]= ℏ e^(+i ϕ) (∂_θ #1 + (i Cos[θ] ∂_ϕ #1)/Sin[θ]) &
```

In[71]:= $\text{Lm} = \left(-\hbar\, e^{-i\phi}\left(\partial_\theta \#- i\,\frac{\text{Cos}[\theta]}{\text{Sin}[\theta]}\,\partial_\phi \#\right)\right)\ \&$

Out[71]= $-\hbar\, e^{-i\phi}\left(\partial_\theta \#1-\frac{i\,\text{Cos}[\theta]\,\partial_\phi \#1}{\text{Sin}[\theta]}\right)\ \&$

In[72]:= $\text{L2} = -\hbar^2\left(\left(\frac{1}{\text{Sin}[\theta]}\,\partial_\theta\,(\text{Sin}[\theta]\,\partial_\theta \#)\right)+\left(\frac{1}{\text{Sin}[\theta]^2}\,\partial_{\phi,\phi}\#\right)\right)\ \&$

Out[72]= $-\hbar^2\left(\frac{\partial_\theta\,(\text{Sin}[\theta]\,\partial_\theta \#1)}{\text{Sin}[\theta]}+\frac{\partial_{\phi,\phi}\#1}{\text{Sin}[\theta]^2}\right)\ \&$

In[73]:= $\text{Lx} = \frac{1}{2}\,(\text{Lp}[\#]+\text{Lm}[\#])\ \&$

Out[73]= $\frac{1}{2}\,(\text{Lp}[\#1]+\text{Lm}[\#1])\ \&$

In[74]:= $\text{Ly} = \frac{1}{2\,i}\,(\text{Lp}[\#]-\text{Lm}[\#])\ \&$

Out[74]= $\frac{\text{Lp}[\#1]-\text{Lm}[\#1]}{2\,i}\ \&$

Define the L^2operator in various ways:

```
In[75]:= L2a = (Lx[Lx[#]] + Ly[Ly[#]] + Lz[Lz[#]]) &;

In[76]:= L2b = (Lp[Lm[#]] + Lz[Lz[#]] - ℏ Lz[#]) &;

In[77]:= L2c = (Lm[Lp[#]] + Lz[Lz[#]] + ℏ Lz[#]) &;
```

Apply the operators on arbitrary (wave)functions:

In[78]:= `Lx[f[θ, ϕ]] // FullSimplify`

Out[78]= $i\,\hbar\,\left(\text{Cos}[\phi]\,\text{Cot}[\theta]\,f^{(0,1)}[\theta,\phi]+\text{Sin}[\phi]\,f^{(1,0)}[\theta,\phi]\right)$

In[79]:= `Ly[f[θ, ϕ]] // FullSimplify`

Out[79]= $i\,\hbar\,\left(\text{Cot}[\theta]\,\text{Sin}[\phi]\,f^{(0,1)}[\theta,\phi]-\text{Cos}[\phi]\,f^{(1,0)}[\theta,\phi]\right)$

In[80]:= `L2a[f[θ, ϕ]] // FullSimplify`

Out[80]= $-\hbar^2\left(\text{Csc}[\theta]^2\,f^{(0,2)}[\theta,\phi]+\text{Cot}[\theta]\,f^{(1,0)}[\theta,\phi]+f^{(2,0)}[\theta,\phi]\right)$

In[81]:= `L2b[f[θ, ϕ]] // FullSimplify`

Out[81]= $-\hbar^2\left(\text{Csc}[\theta]^2\,f^{(0,2)}[\theta,\phi]+\text{Cot}[\theta]\,f^{(1,0)}[\theta,\phi]+f^{(2,0)}[\theta,\phi]\right)$

In[82]:= `L2c[f[θ, ϕ]] // FullSimplify`

Out[82]= $-\hbar^2\left(\text{Csc}[\theta]^2\,f^{(0,2)}[\theta,\phi]+\text{Cot}[\theta]\,f^{(1,0)}[\theta,\phi]+f^{(2,0)}[\theta,\phi]\right)$

In[83]:= `L2[f[θ, ϕ]] // FullSimplify`

Out[83]= $-\hbar^2\left(\text{Csc}[\theta]^2\,f^{(0,2)}[\theta,\phi]+\text{Cot}[\theta]\,f^{(1,0)}[\theta,\phi]+f^{(2,0)}[\theta,\phi]\right)$

Commutation relations of Angular Momentum Operators

In[84]:= `Lzs = (Lp[Lm[#]] - Lm[Lp[#]]) & // FullSimplify`

Out[84]= `Lp[Lm[#1]] - Lm[Lp[#1]] &`

In[85]:= `Lzs[f[θ, ϕ]] // FullSimplify`

Out[85]= $-2\,i\,\hbar^2\,f^{(0,1)}[\theta,\phi]$

```
In[86]:= 2 ℏ Lz[f[θ, ϕ]]
```

Out[86]= $-2\, i\, \hbar^2\, f^{(0,1)}[\theta, \phi]$

```
In[87]:= Lzs[f[θ, ϕ]] == 2 ℏ Lz[f[θ, ϕ]] // FullSimplify
```

Out[87]= True

```
In[88]:= Lzss = (Lx[Ly[#]] - Ly[Lx[#]]) &;
```

```
In[89]:= Lzss[f[θ, ϕ]] // FullSimplify
```

Out[89]= $\hbar^2\, f^{(0,1)}[\theta, \phi]$

```
In[90]:= i ℏ Lz[f[θ, ϕ]]
```

Out[90]= $\hbar^2\, f^{(0,1)}[\theta, \phi]$

```
In[91]:= Lzss[f[θ, ϕ]] == i ℏ Lz[f[θ, ϕ]] // FullSimplify
```

Out[91]= True

Calculate the expectation values

$< Y_1{}^1 \,|\, L_+ L_- Y_1{}^1 >$

In[92]:= $\int_0^{2\pi}\int_0^{\pi}$ SphericalHarmonicY[1, 1, θ, ϕ]* Lp[Lm[SphericalHarmonicY[1, 1, θ, ϕ]]] Sin[θ] ⅆθ ⅆϕ

Out[92]= $2\,\hbar^2$

$< L_- Y_1{}^1 \,|\, L_- Y_1{}^1 >$

In[93]:= $\int_0^{2\pi}\int_0^{\pi}$ (Lm[SphericalHarmonicY[1, 1, θ, ϕ]])* Lm[SphericalHarmonicY[1, 1, θ, ϕ]] Sin[θ] ⅆθ ⅆϕ

Out[93]= $2\,\hbar^2$

Apply Angular momentum operators on Spherical harmonics

```
In[94]:= ψ[θ_, ϕ_] = SphericalHarmonicY[1, 1, θ, ϕ]
```

Out[94]= $-\frac{1}{2} e^{i\phi} \sqrt{\frac{3}{2\pi}}\, \mathrm{Sin}[\theta]$

In[95]:= $\int_0^{2\pi}\int_0^{\pi}$ ψ[θ, ϕ]* ψ[θ, ϕ] Sin[θ] ⅆθ ⅆϕ

Out[95]= 1

In[96]:= $\int_0^{2\pi}\int_0^{\pi}$ ψ[θ, ϕ]* Lp[Lm[ψ[θ, ϕ]]] Sin[θ] ⅆθ ⅆϕ

Out[96]= $2\,\hbar^2$

In[97]:= $\int_0^{2\pi}\int_0^{\pi}$ (Lm[ψ[θ, ϕ]])* Lm[ψ[θ, ϕ]] Sin[θ] ⅆθ ⅆϕ

Out[97]= $2\,\hbar^2$

In[98]:= $\int_0^{2\pi}\int_0^{\pi}$ ψ[θ, ϕ]* L2[ψ[θ, ϕ]] Sin[θ] ⅆθ ⅆϕ

Out[98]= $2\,\hbar^2$

In[99]:= $\psi_{1\_,m\_}$[θ_, ϕ_] = SphericalHarmonicY[l, m, θ, ϕ];

In[100]:= $\int_0^{2\pi}\int_0^{\pi}$ $\psi_{3,-2}$[θ, ϕ]* L2[$\psi_{3,-2}$[θ, ϕ]] Sin[θ] ⅆθ ⅆϕ

Out[100]= $12\,\hbar^2$

In[101]:= $\int_0^{2\pi}\int_0^{\pi} \psi_{3,-2}[\theta, \phi]^* \, \text{L2a}[\psi_{3,-2}[\theta, \phi]] \, \text{Sin}[\theta] \, d\theta \, d\phi$

Out[101]= $12\,\hbar^2$

In[102]:= $\int_0^{2\pi}\int_0^{\pi} \psi_{3,-2}[\theta, \phi]^* \, \text{Lz}[\psi_{3,-2}[\theta, \phi]] \, \text{Sin}[\theta] \, d\theta \, d\phi$

Out[102]= $-2\,\hbar$

Summary

- We made a start with Mathematica skills and applications in Applied Physics.
- We did set out a line through the Applied Physics curriculum, with a balanced set-up.
- We tried to balance the learning skills with the useful applications in the fundamental physics courses.
- No response yet from the Electrodynamics experience.
- Preparation of implementation in Quantum course.

Conclusions

- We did our first small steps.
- No conclusive remarks yet.
- Some items for discussion...

Items for discussion

1. When to start with Computer Algebra?
2. How to balance the skills training and motivating applications?
3. How to use Mathematica in exams?